

EBNF of SOL

program → *func-decl*
func-decl → **func** **id** (*decl-list-opt*) : *domain* *type-sect-opt* *var-sect-opt* *const-sect-opt* *func-list-opt* *func-body*
decl-list-opt → { *decl* ; }
decl → *id-list* : *domain* ;
id-list → **id** { , **id** }
domain → *atomic-domain* | *struct-domain* | *vector-domain* | **id**
atomic-domain → **char** | **int** | **real** | **string** | **bool**
struct-domain → **struct** (*decl-list*)
vector-domain → **vector** [**intconst**] **of** *domain*
type-sect-opt → [**type** { *decl* }⁺]
var-sect-opt → [**var** { *decl* }⁺]
const-sect-opt → [**const** { *const-decl* }⁺]
const-decl → *decl* = *expr* ;
func-list-opt → { *func-decl* }
func-body → **begin** **id** *stat-list* **end** **id**
stat-list → { *stat* ; }⁺
stat → *assign-stat* | *if-stat* | *while-stat* | *for-stat* | *foreach-stat* | *return-stat* | *read-stat* | *write-stat*
assign-stat → *left-hand-side* = *expr*
left-hand-side → **id** { (. **id** | [*expr*]) }
if-stat → **if** *expr* **then** *stat-list* { **elsif** *expr* **then** *stat-list* } [**else** *stat-list*] **endif**
while-stat → **while** *expr* **do** *stat-list* **endwhile**
for-stat → **for** **id** = *expr* **to** *expr* **do** *stat-list* **endfor**
foreach-stat → **foreach** **id** **in** *expr* **do** *stat-list* **endforeach**
return-stat → **return** *expr*
read-stat → **read** *specifier-opt* **id**
specifier-opt → [[*expr*]]
write-stat → **write** *specifier-opt* *expr*
expr → *bool-term* { *bool-op* *bool-term* }
bool-op → **and** | **or**
bool-term → *rel-term* [*rel-op* *rel-term*]
rel-op → == | != | > | >= | < | <= | **in**
rel-term → *low-term* { *low-bin-op* *low-term* }
low-bin-op → + | -
low-term → *factor* { *high-bin-op* *factor* }
high-bin-op → * | /
factor → *unary-op* *factor* | (*expr*) | *left-hand-side* | *atomic-const* | *instance-construction* |
 func-call | *cond-expr* | *built-in-call* | *dynamic-input*
unary-op → - | **not** | *dynamic-output*
atomic-const → **charconst** | **intconst** | **realconst** | **strconst** | **boolconst**
instance-construction → *struct-construction* | *vector-construction*
struct-construction → **struct** (*expr-list*)
expr-list → *expr* { , *expr* }
vector-construction → **vector** (*expr-list*)
func-call → **id** ([*expr-list*])
cond-expr → **if** *expr* **then** *expr* { **elsif** *expr* **then** *expr* } **else** *expr* **endif**
built-in-call → (**toint** | **toreal**) (*expr*)
dynamic-input → **rd** *specifier-opt* *domain*
dynamic-output → **wr** *specifier-opt*